

Erasing propositions and homotopy type theory

Gabe Dijkstra

June 26, 2013

Contents

1	Introduction	5
2	Homotopy type theory	7
3	Erasing propositions	9
3.1	Propositions	10
3.1.1	Bove-Capretta method	10
3.2	The <i>Prop</i> universe in Coq	11
3.2.1	Singleton elimination and homotopy type theory	12
3.2.2	Quicksort example	13
3.2.3	Impredicativity	13
3.3	Irrelevance in Agda	14
3.3.1	Quicksort example	16
3.4	Collapsible families	16
3.4.1	Erasing concretely collapsible families	18
3.4.2	Quicksort example	18
3.5	Internalising collapsibility	19
3.6	Internalising the collapsibility optimisation	20
3.6.1	Time complexity issues	21
3.7	Indexed <i>h</i> -propositions and homotopy type theory	22
3.7.1	Internally optimising <i>h</i> -propositions	24
3.8	Conclusion and future work	24
4	Applications of homotopy type theory	25

Chapter 1

Introduction

Chapter 2

Homotopy type theory

Chapter 3

Erasing propositions

When writing certified programs in a dependently typed setting, we can conceptually distinguish between the *program* parts and the *proof* (of correctness) parts. These are sometimes also referred to as the informative and logical parts, respectively. In practice, these two seemingly separate concerns are often intertwined. Consider for example the sorting of lists of naturals: given some predicate $isSorted : List \mathbb{N} \rightarrow List \mathbb{N} \rightarrow Type$ that tells us whether the second list is a sorted permutation of the first one, we can write a term of the following type:

$$sort : (xs : List \mathbb{N}) \rightarrow \Sigma (ys : List \mathbb{N}) (isSorted xs ys)$$

To implement such a function, we need to provide for every list a sorted list along with a proof that this is indeed a sorted version of the input list. At runtime the type checking has been done, hence the proof of correctness has already been verified: we want to *erase* the logical parts.

Types such as $isSorted xs ys$ are purely logical: we care more about the presence of an inhabitant than what kind of inhabitant we exactly have at our disposal. In section 3.1 we will give more examples of such types, called *propositions*, and how they can occur in various places in certified programs. In sections 3.2 and 3.3 we review the methods Coq and Agda provide us to annotate parts of our program as being propositions. Section 3.4 reviews the concept of *collapsible families* and how we can automatically detect whether a type is a proposition, instead of annotating them ourselves. In section 3.5 we internalise the concept of collapsible families and try to do the same with the optimisation in section 3.6. The internalised version of collapsibility looks like an indexed version of the concept of *h*-propositions. In section 3.7 we investigate if we can use this to devise an optimisation akin to the optimisation based on collapsibility.

3.1 Propositions

In the *sort* example, the logical part *isSorted xs ys* occurs in the result as part of a Σ -type. This means we can separate the proof of correctness from the sorting itself, i.e. we can write a function *sort'* : *List* \mathbb{N} \rightarrow *List* \mathbb{N} and a proof of the following:

$$\text{sortCorrect} : (xs : \text{List } \mathbb{N}) \rightarrow \text{isSorted } xs \ (\text{sort}' \ xs)$$

The logical part here asserts properties of the *result* of the computation. If we instead have assertions on our *input*, we cannot decouple this from the rest of the function as easily as, if it is at all possible. For example, suppose we have a function, safely selecting the *n*-th element of a list:

$$\text{elem} : (A : \text{Type}) \ (xs : \text{List } A) \ (i : \mathbb{N}) \rightarrow i < \text{length } xs \rightarrow A$$

If we were to write *elem* without the bounds check $i < \text{length } xs$, we would get a partial function. Since we can only define total functions in our type theory, we cannot write such a function. However, at run-time, carrying these proofs around makes no sense: type checking has already shown that all calls to *elem* are safe and the proofs do not influence the outcome of *elem*. We want to erase terms of types such as $i < \text{length } xs$, if we have established that they do not influence the run-time computational behaviour of our functions.

3.1.1 Bove-Capretta method

The *elem* example showed us how we can use propositions to write functions that would otherwise be partial, by asserting properties of the input. The Bove-Capretta method (Bove and Capretta, 2005) generalises this and more: it provides us with a way to transform any (possibly partial) function defined by general recursion into a total, structurally recursive one. The quintessential example of a definition that is not structurally recursive is *quicksort*¹ :

$$\begin{aligned} qs &: \text{List } \mathbb{N} \rightarrow \text{List } \mathbb{N} \\ qs [] &= [] \\ qs (x :: xs) &= qs (\text{filter } (gt \ x) \ xs) \uplus x :: qs (\text{filter } (le \ x) \ xs) \end{aligned}$$

The recursive calls are done on *filter* (*gt* *x*) *xs* and *filter* (*le* *x*) *xs* instead of just *xs*, hence *qs* is not structurally recursive. To solve this problem, we create an inductive family describing the call graphs of the original function for every input. Since we can only construct finite values, being able to produce such a call graph essentially means that the function terminates for that input. We can then write a new function that structurally recurses on the call graph. In our quicksort case we get the following inductive family:

¹In most implementations of functional languages, this definition will not have the same space complexity as the usual in-place version. We are more interested in this function as an example of non-structural recursion and are not too concerned with its complexity.

```

data qsAcc : List ℕ → Set where
  qsAccNil   : qsAcc []
  qsAccCons : (x : ℕ) (xs : List ℕ)
               (h1 : qsAcc (filter (gt x) xs))
               (h2 : qsAcc (filter (le x) xs))
               → qsAcc (x :: xs)

```

with the following function definition²

```

qs : (xs : List ℕ) → qsAcc xs → List ℕ
qs .nil   qsAccNil           = []
qs .cons (qsAccCons x xs h1 h2) = qs (filter (gt x) xs) h1 ++
                                     x :: qs (filter (le x) xs) h2

```

Pattern matching on the *qsAcc xs* argument gives us a structurally recursive version of *qs*. Just as with the *elem* example, we need information from the proof to be able to write this definition in our type theory. In the case of *elem*, we need the proof of $i < \text{length } xs$ to deal with the (impossible) case where *xs* is empty. In the *qs* case, we need *qsAcc xs* to guide the recursion. Even though we actually pattern match on *qsAcc xs* and it therefore seemingly influences the computational behaviour of the function, erasing this argument yields the original *qs* definition.

3.2 The Prop universe in Coq

In Coq we have the *Prop* universe, apart from the *Set* universe. Both universes are base sorts of the hierarchy of sorts, *Type*, i.e. *Prop* : *Type* (1), *Set* : *Type* (1) and for every i , *Type* (i) : *Type* (i + 1). As the name suggests, by defining a type to be of sort *Prop*, we “annotate” it to be a logical type, a proposition. Explicitly marking the logical parts like this, makes the development easier to read and understand. More importantly, the extraction mechanism (Letouzey, 2003) now knows what parts are supposed to be logical, hence what parts are to be erased.

In the *sort* example, we would define *isSorted* to be a family of *Props* indexed by *List* ℕ. For the Σ -type, Coq provides two options: *sig* and *ex*, defined as follows:

```

Inductive sig (A : Type) (P : A → Prop) : Type :=
  exist : ∀ x : A, P x → sig P
Inductive ex (A : Type) (P : A → Prop) : Prop :=
  ex_intro : ∀ x : A, P x → ex P

```

As can be seen above, *sig* differs from *ex* in that the latter is completely logical, whereas *sig* has one informative and one logical field and in its entirety is informative. Since we are interested in the *list* ℕ part of the Σ -type that is the result type of *sort*, but not the *isSorted* part, we choose the *sig* version.

²This definition uses dependent pattern matching (Coquand, 1992), but can be rewritten directly using the elimination operators instead. The important thing here is to notice that we are eliminating the *qsAcc xs* argument.

The extracted version of *sig* consists of a single constructor *exist*, with a single field of type *A*. Since this is isomorphic to the type *A* itself, Coq optimises this away during extraction. This means $\text{sort} : (xs : \text{List } \mathbb{N}) \rightarrow \Sigma (ys : \text{List } \mathbb{N}) (\text{isSorted } xs \ ys)$ gets extracted to a function $\text{sort}' : \text{List } \mathbb{N} \rightarrow \text{List } \mathbb{N}$.

When erasing all the *Prop* parts from our program, we do want to retain the computational behaviour of the remaining parts. Every function that takes an argument of sort *Prop*, but whose result type is not in *Prop*, needs to be invariant under choice of inhabitant for the *Prop* argument. To force this property, Coq restricts the things we can eliminate a *Prop* into. The general rule is that pattern matching on something of sort *Prop* is allowed if the result type of the function happens to be in *Prop*.

3.2.1 Singleton elimination and homotopy type theory

There are exceptions to this rule: if the argument we are pattern matching on happens to be an *empty* or *singleton definition* of sort *Prop*, we may also eliminate into *Type*. An empty definition is an inductive definition without any constructors. A singleton definition is an inductive definition with precisely one constructor, whose fields are all in *Prop*. Examples of such singleton definitions are conjunction on *Prop* (\wedge) and the accessibility predicate *Acc* used to define functions using well-founded recursion.

Another important example of singleton elimination is elimination on Coq's equality *eq* (where $a = b$ is special notation for $\text{eq } a \ b$), which is defined to be in *Prop*. The inductive family *eq* is defined in the same way as we have defined identity types, hence it is a singleton definition, amenable to singleton elimination. Consider for example the *transport* function:

Definition $\text{transport} : \forall A, \forall (P : A \rightarrow \text{Type}),$
 $\forall (x \ y : A),$
 $\forall (\text{path} : x = y),$
 $P \ x \rightarrow P \ y .$

Singleton elimination allows us to pattern match on *path* and eliminate into something of sort *Type*. In the extracted version, the *path* argument gets erased and the $P \ x$ argument is returned. In homotopy type theory, we know that the identity types need not be singletons and can have other inhabitants than just the canonical *refl*, so throwing away the identity proof is not correct. As has been discovered by Michael Shulman³, singleton elimination leads to some sort of inconsistency, if we assume the univalence axiom: we can construct a value $x : \text{bool}$ such that we can prove $x = \text{false}$, even though in the extracted version x normalises to *true*. Assuming univalence, we have two distinct proofs of $\text{bool} = \text{bool}$, namely *refl* and the proof we get from applying univalence to the isomorphism $\text{not} : \text{bool} \rightarrow \text{bool}$. Transporting a value along a path we have obtained from using univalence, is the same as applying the isomorphism. Defining x to be *true* transported along the path obtained from applying univalence to the isomorphism *not*, yields something that is propositionally equal to *false*.

³<http://homotopytypetheory.org/2012/01/22/univalence-versus-extraction/>

If we extract the development, we get a definition of x that ignores the proof of $bool = bool$ and just returns *true*.

In other words, Coq does not enforce or check proof irrelevance of the types we define to be of sort *Prop*, which internally is fine: it does not allow us to derive falsity using this fact. The extraction mechanism however, does assume that everything admits proof irrelevance. The combination of this along with singleton elimination means that we can prove properties about our programs that no longer hold in the extracted version. It also goes to show that the design decision to define the identity types to be in *Prop* is not compatible with homotopy type theory.

3.2.2 Quicksort example

In the case of qs defined using the Bove-Capretta method, we actually want to pattern match on the logical part: $qsAcc\ xs$. Coq does not allow this if we define the family $qsAcc$ to be in *Prop*. However, we can do the pattern matching “manually”, as described in Bertot and Castéran (2004). We know that we have exactly one inhabitant of $qsAcc\ xs$ for each xs , as they represent the call graph of qs for the input xs , and the pattern matches of the original definition do not overlap, hence each xs has a unique call graph. We can therefore easily define and prove the following inversion theorems, that roughly look as follows:

$$\begin{aligned} qsAccInv_0 &: (x : \mathbb{N}) (xs : List\ \mathbb{N}) (qsAcc\ (x :: xs)) \rightarrow qsAcc\ (filter\ (le\ x)\ xs) \\ qsAccInv_1 &: (x : \mathbb{N}) (xs : List\ \mathbb{N}) (qsAcc\ (x :: xs)) \rightarrow qsAcc\ (filter\ (gt\ x)\ xs) \end{aligned}$$

We define the function qs just as we originally intended to and add the $qsAcc\ xs$ argument to every pattern match. We then call the inversion theorems for the appropriate recursive calls. Coq still notices that there is a decreasing argument, namely $qsAcc\ xs$. If we follow this approach, we can define $qsAcc$ to be a family in *Prop* and recover the original qs definition without the $qsAcc\ xs$ argument using extraction.

In the case of partial functions, we still have to add the missing pattern matches and define impossibility theorems: if we reach that pattern match and we have a proof of our Bove-Capretta predicate for that particular pattern match, we can prove falsity, hence we can use *False_rect* to deal with the missing pattern match.

3.2.3 Impredicativity

So far we have seen how *Prop* differs from *Set* with respect to its restricted elimination rules and its erasure during extraction, but *Prop* has another property that sets it apart from *Set*: *impredicativity*. Impredicativity means that we are able to quantify over something which contains the thing currently being defined. In set theory unrestricted use of this principle leads us to being able to construct Russell’s paradox: the set $R = \{x | x \in x\}$ is an impredicative definition, we quantify over x , while we are also defining x . Using this definition we

can prove that $R \in R$ if and only if $R \notin R$. In type theory, an analogous paradox, Girard's paradox, arises if we allow for impredicativity via the $Type : Type$ rule. However, impredicative definitions are sometimes very useful and benign, in particular when dealing with propositions: we want to be able to write propositions that quantify over propositions, for example:

Definition *demorgan* : $Prop := \forall P Q : Prop,$
 $\sim(P \wedge Q) \rightarrow \sim P \vee \sim Q$.

Coq allows for such definitions as the restrictions on *Prop* prevent us from constructing paradoxes such as Girard's.

3.3 Irrelevance in Agda

In Coq, we put the annotations of something being a proposition in the definition of our inductive type, by defining it to be of sort *Prop*. With Agda's irrelevance mechanism, we instead put the annotations at the places we *use* the proposition, by placing a dot in front of the corresponding type. For example, the type of the *elem* becomes:

$elem : (A : Type) (xs : List A) (i : \mathbb{N}) \rightarrow .(i < length\ xs) \rightarrow A$

We can also mark fields of a record to be irrelevant. In the case of *sort*, we want something similar to the *sig* type from Coq, where second field of the Σ -type is deemed irrelevant. In Agda this can be done as follows:

record Σ_{irr} ($A : Type$) ($B : A \rightarrow Type$) : *Type* **where**
constructor $_, -$
field
 $fst : A$
 $.snd : B\ fst$

To ensure that irrelevant arguments are indeed irrelevant to the computation at hand, Agda has several criteria that it checks. First of all, no pattern matching may be performed on irrelevant arguments, just as is the case with *Prop*. (However, the absurd pattern may be used, if applicable.) Contrary to Coq, singleton elimination is not allowed. Secondly, we need to ascertain that the annotations are preserved: irrelevant arguments may only be passed on to irrelevant contexts. This prevents us from writing a function of type $(A : Type) \rightarrow .A \rightarrow A$.

Another, more important, difference with *Prop* is that irrelevant arguments are ignored by the type checker when checking equality of terms. This can be done safely, even though the terms at hand may in fact be definitionally different, as we never need to appeal to the structure of the value: we cannot pattern match on it. The only thing that we can do with irrelevant arguments is either ignore them or pass them around to other irrelevant contexts.

The reason why the type checker ignoring irrelevant arguments is important, is that it allows us to prove properties about irrelevant arguments in Agda, internally. For example: any function out of an irrelevant type is constant:

$$\begin{aligned} \text{irrelevantConstantFunction} &: \{A : \text{Type}\} \{B : \text{Type}\} \\ &\rightarrow (f : A \rightarrow B) \rightarrow (x \ y : A) \rightarrow f \ x \equiv f \ y \\ \text{irrelevantConstantFunction } f \ x \ y &= \text{refl} \end{aligned}$$

There is no need to use the congruence rule for \equiv , since the x and y are ignored when the type checker compares $f \ x$ to $f \ y$, when type checking the refl . The result can be easily generalised to dependent functions:

$$\begin{aligned} \text{irrelevantConstantDepFunction} &: \{A : \text{Type}\} \{B : A \rightarrow \text{Type}\} \\ &\rightarrow (f : A \rightarrow B) \rightarrow (x \ y : A) \rightarrow f \ x \equiv f \ y \\ \text{irrelevantConstantDepFunction } f \ x \ y &= \text{refl} \end{aligned}$$

Note that we do not only annotate $(x : A)$ with a dot, but also occurrence of A in the type $B : A \rightarrow \text{Type}$, otherwise we are not allowed to write $B \ x$ as we would use an irrelevant argument in a relevant context. When checking $\text{irrelevantConstantDepFunction}$, the term $f \ x \equiv f \ y$ type checks, without having to transport one value along some path, because the types $B \ x$ and $B \ y$ are regarded as definitionally equal by the type checking, ignoring the x and y . Just as before, there is no need to use the (dependent) congruence rule; a refl suffices.

We would also like to show that we have proof irrelevance for irrelevant arguments, i.e. we want to prove the following:

$$\text{irrelevantProofIrrelevance} : \{A : \text{Type}\} . (x \ y : A) \rightarrow x \equiv y$$

Agda does not accept this, because the term $x \equiv y$ uses irrelevant arguments in a relevant context: $x \equiv y$. If we instead package the irrelevant arguments in an inductive type, we can prove that the two values of the packaged type are propositionally equal. Consider the following record type with only one irrelevant field:

```
record Squash (A : Type) : Type where
  constructor squash
  field
    .proof : A
```

Using this type, we can now formulate the proof irrelevance principle for irrelevant arguments and prove it:

$$\begin{aligned} \text{squashProofIrrelevance} &: \{A : \text{Type}\} (x \ y : \text{Squash } A) \rightarrow x \equiv y \\ \text{squashProofIrrelevance } x \ y &= \text{refl} \end{aligned}$$

The name “squash type” comes from Nuprl (Constable et al., 1986): one takes a type and identifies (or “squashes”) all its inhabitants into one unique (up to propositional equality) inhabitant. In homotopy type theory the process of

squashing a type is called (-1) -truncation and can also be achieved by defining the following higher inductive type:

```
data  $(-1)$ -truncation  $(A : Type) : Type$  where
  inhab : A
  all-paths :  $(x\ y : A) \rightarrow x \equiv y$ 
```

3.3.1 Quicksort example

If we want to mark the *qsAcc xs* argument of the *qs* function as irrelevant, we run into the same problems as we did when we tried to define *qsAcc* as a family in *Prop*: we can no longer pattern match on it. In Coq, we did have a way around this, by using inversion and impossibility theorems to do the pattern matching “manually”. However, if we try such an approach in Agda, its termination checker cannot see that *qsAcc xs* is indeed a decreasing argument and refuses the definition.

3.4 Collapsible families

The approaches we have seen so far let the user indicate what parts of the program are the logical parts and are amenable for erasure. Brady et al. (2004) show that we can let the compiler figure that out by itself instead. The authors propose a series of optimisations for the Epigram system, based on the observation that one often has a lot of redundancy in well-typed terms. If it is the case that one part of a term has to be definitionally equal to another part in order to be well-typed, we can leave out (presuppose) the latter part if we have already established that the term is well-typed.

The authors describe their optimisations in the context of Epigram. In this system, the user writes programs in a high-level language that gets elaborated to programs in a small type theory language. This has the advantage that if we can describe a translation for high-level features, such as dependent pattern matching, to a simple core type theory, the metatheory becomes a lot simpler. The smaller type theory also allows us to specify optimisations more easily, because we do not have to deal with the more intricate, high-level features.

As such, the only things we need to look at, if our goal is to optimise a certain inductive family, are its constructors and its elimination principle. Going back to the *elem* example, we had the $i < \text{length } xs$ argument. The smaller-than relation can be defined as the following inductive family (in Agda syntax):

```
data  $_ < _ : \mathbb{N} \rightarrow \mathbb{N} \rightarrow Type$  where
  ltZ :  $(y : \mathbb{N}) \rightarrow Z < S\ y$ 
  ltS :  $(x\ y : \mathbb{N}) \rightarrow x < y \rightarrow S\ x < S\ y$ 
```

with elimination operator

$$\begin{aligned}
<-elim : & (P : (x \ y : \mathbb{N}) \rightarrow x < y \rightarrow Type) \\
& (m_Z : (y : \mathbb{N}) \rightarrow P \ 0 \ (S \ y) \ (ltZ \ y)) \\
& (m_S : (x \ y : \mathbb{N}) \rightarrow (pf : x < y) \rightarrow P \ x \ y \ pf \rightarrow P \ (S \ x) \ (S \ y) \ (ltS \ x \ y \ pf)) \\
& (x \ y : \mathbb{N}) \\
& (pf : x < y) \\
& \rightarrow P \ x \ y \ pf
\end{aligned}$$

and computation rules

$$\begin{aligned}
<-elim \ P \ m_Z \ m_S \ 0 \ (S \ y) \ (ltZ \ y) & \mapsto m_Z \ y \\
<-elim \ P \ m_Z \ m_S \ (S \ x) \ (S \ y) \ (ltS \ x \ y \ pf) & \mapsto m_S \ x \ y \ pf \ (<-elim \ P \ m_Z \ m_S \ x \ y \ pf)
\end{aligned}$$

If we look at the computation rules, we see that we can presuppose several things. The first rule has a repeated occurrence of y , so we can presuppose the latter occurrence, the argument of the constructor. In the second rule, the same can be done for x and y . The pf argument can also be erased, as it is never inspected: the only way to inspect pf is via another call the $<-elim$, so by induction it is never inspected. Another thing we observe is that the pattern matches on the indices are disjoint, so we can presuppose the entire target: everything can be recovered from the indices given to the call of $<-elim$.

We have to be careful when making assumptions about values, given their indices. Suppose we have written a function that takes $p : 1 < 1$ as an argument and contains a call to $<-elim$ on p . If we look at the pattern matches on the indices, we may be led to believe that p is of form $ltS \ 0 \ 0 \ p'$ for some $p' : 0 < 0$ and reduce accordingly. The presupposing only works for *canonical* values, hence we restrict our optimisations to the run-time (evaluation in the empty context), as we know we do not perform reductions under binders in that case and every value is canonical after reduction. The property that every term that is well-typed in the empty context, reduces to a canonical form is called *adequacy* and is a property that is satisfied by Martin-Löf's type theory.

The family $<-elim$ has the property that for indices $x \ y : \mathbb{N}$, its inhabitants $p : x < y$ are uniquely determined by these indices. To be more precise, the following is satisfied: for all $x \ y : \mathbb{N}$, $\vdash p \ q : x < y$ implies $\vdash p \triangleq q$. Families $D : I_0 \rightarrow \dots \rightarrow I_n \rightarrow Type$ such as $<-elim$ are called *collapsible* if they satisfy that for every $i_0 : I_0, \dots, i_n : I_n$, if $\vdash p \ q : D \ i_0 \ \dots \ i_n$, then $\vdash p \triangleq q$.

Checking collapsibility of an inductive family is undecidable in general. This can be seen by reducing it to the type inhabitation problem: consider the type $\top + A$. This type is collapsible if and only if A is uninhabited, hence determining with being able to decide collapsibility means we can decide type inhabitation as well. As such, we limit ourselves to a subset that we can recognise, called *concretely* collapsible families. A family $D : I_0 \rightarrow \dots \rightarrow I_n \rightarrow Type$ is concretely collapsible if satisfies the following two properties:

- If we have $\vdash x : D \ i_0 \ \dots \ i_n$, for some $i_0 : I_0, \dots, i_n : I_n$, then we can recover its constructor tag by pattern matching on the indices.
- All the non-recursive arguments to the constructors of D can be recovered by pattern matching on the indices.

Note that the first property makes sense because we only have to deal with canonical terms, due to the adequacy property. Checking whether this first property holds can be done by checking whether the indices of the constructors, viewed as patterns, are disjoint. The second property can be checked by pattern matching on the indices of every constructor and checking whether the non-recursive arguments occur as pattern variables.

3.4.1 Erasing concretely collapsible families

If D is a collapsible family, then its elimination operator $D - \text{elim}$ is constant in its target, if we fix the indices. This seems to indicate that there might be a possibility to erase the target altogether. Nevertheless, D might have constructors with non-recursive arguments giving us information. Concretely collapsible families satisfy the property that this kind of information can be recovered from the indices, so we can get away with erasing the entire target. Being concretely collapsible means that we have a function at the meta-level (or implementation level) from the indices to the non-recursive, relevant parts of the target. Since this is done by pattern matching on the fully evaluated indices, recovering these parts takes an amount of time that is constant in the size of the given indices. Even though this sounds promising, the complexity of patterns does influence this constant, e.g. the more deeply nested the patterns are, the higher the constant. We now also need the indices to be fully evaluated when eliminating a particular inductive family, whereas that previously might not have been needed. The optimisation is therefore one that gives our dependently typed programs a better space complexity, but not necessarily a better time complexity.

3.4.2 Quicksort example

The accessibility predicates $qsAcc$ form a collapsible family. The pattern matches on the indices in the computation rules for $qsAcc$ are the same pattern matches as those of the original qs definition. There are no overlapping patterns in the original definition, so we can indeed recover the constructor tags from the indices. Also, the non-recursive arguments of $qsAcc$ are precisely those given as indices, hence $qsAcc$ is indeed a (concretely) collapsible family. By the same reasoning, any Bove-Capretta predicate is concretely collapsible, given that the original definition we derived the predicate from, has disjoint pattern matches.

The most important aspect of the collapsibility optimisation is that we have established that everything we need from the value that is to be erased, can be (cheaply) *recovered* from its indices passed to the call to its elimination operator. This means that we have no restrictions on the elimination of collapsible families: we can just write our definition of qs by pattern matching on the $qsAcc\ xs$ argument. At run-time, the $qsAcc\ xs$ argument has been erased and the relevant parts are recovered from the indices.

3.5 Internalising collapsibility

Checking whether an inductive family is concretely collapsible is something that can be easily done automatically, as opposed to determining collapsibility in general, which is undecidable. In this section we investigate if we can formulate an internal version of collapsibility, enabling the user to give a proof that a certain family is collapsible, if the compiler fails to notice so itself.

Recall the definition of a collapsible family⁴: given an inductive family D indexed by the type I , D is collapsible if for every index $i : I$ and terms x, y , the following holds:

$$\vdash x, y : D \ i \text{ implies } \vdash x \triangleq y$$

This definition makes use of definitional equality. Since we are working with an intensional type theory, we do not have the *equality reflection rule* at our disposal: there is no rule that tells us that propositional equality implies definitional equality. This might lead us to think that internalising the above definition will not work, as we seemingly cannot say anything about definitional equality from within Martin-Löf's type theory. Let us consider the following variation: for all terms x, y there exists a term p such that

$$\vdash x, y : D \ i \text{ implies } \vdash p : x \equiv y$$

Since Martin-Löf's type theory satisfies the canonicity property, any term p such that $\vdash p : x \equiv y$ reduces to *refl*. The only way for the term to type check, is if $x \triangleq y$, hence in the empty context the equality reflection rule does hold. The converse is also true: definitional equality implies of x and y that $\vdash \text{refl} : x \equiv y$ type checks, hence the latter definition is equal to the original definition of collapsibility.

The variation given above is still not a statement that we can directly prove internally: we need to internalise the implication and replace it by the function space. Doing so yields the following definition: there exists a term p such that:

$$\vdash p : (i : I) \rightarrow (x \ y : D \ i) \rightarrow x \equiv y$$

Or, written as a function in Agda:

$$\begin{aligned} \text{isInternallyCollapsible} &: (I : \text{Type}) (A : I \rightarrow \text{Type}) \rightarrow \text{Type} \\ \text{isInternallyCollapsible } I \ A &= (i : I) \rightarrow (x \ y : A \ i) \rightarrow x \equiv y \end{aligned}$$

We will refer to this definition as *internal collapsibility*. It is easy to see that every internally collapsible family is also collapsible, by canonicity and the fact that *refl* implies definitional equality. However, internally collapsible families do

⁴The definition we originally gave allowed for an arbitrary number of indices. In the following sections we will limit ourselves to the case where we have only one index for presentation purposes. All the results given can be easily generalised to allow more indices.

differ from collapsible families as can be seen by considering D to be the family Id . By canonicity we have that for any $A : Type$, $x, y : A$, a term p satisfying $\vdash p : Id\ A\ x\ y$ necessarily reduces to $refl$. This means that Id is a collapsible family. In contrast, Id does not satisfy the internalised condition given above, since this then boils down to the uniqueness of identity proofs principle, which does not hold, as we have discussed.

3.6 Internalising the collapsibility optimisation

In section 3.4.1 we saw how concretely collapsible families can be erased, since all we want to know about the inhabitants can be recovered from its indices. In this section we will try to uncover a similar optimisation for internally collapsible families.

We cannot simply erase the internally collapsible arguments from the function we want to optimise, e.g. given a function $f : (i : I) \rightarrow (x : D\ i) \rightarrow \tau$, we generally cannot produce a function $\tilde{f} : (i : I) \rightarrow \tau$, since we sometimes need the $x : D\ i$ in order for the function to typecheck. However, we can use Agda's irrelevance mechanism to instead generate a function in which the collapsible argument is marked as irrelevant, i.e. we want to write the following function (for the non-dependent case):

```
optimiseFunction :
  (I : Type) (A : I → Type) (B : Type)
  (isInternallyCollapsible I A)
  (f : (i : I) → A i → B)
  → ((i : I) → .(A i) → B)
```

Along with such a function, we should also give a proof that the generated function is equal to the original one in the following sense:

```
optimiseFunctionCorrect :
  (I : Type) (D : I → Type) (B : Type)
  (pf : isInternallyCollapsible I D)
  (f : (i : I) → D i → B)
  (i : I) (x : D i)
  → optimiseFunction I D B pf f i x ≡ f i x
```

If we set out to write the function *optimiseFunction*, after having introduced all the variables, our goal is to produce something of type B . This can be done by using the function f , but then we need a $i : I$ and something of type $D\ i$. We have both, however the $D\ i$ we have is marked as irrelevant, so it may only be passed along to irrelevant contexts, which the function f does not provide, so we cannot use that one. We need to find another way to produce an $D\ i$. We might try to extract it from the proof of *isInternallyCollapsible I D*, but this proof only tells us how the inhabitants of every $D\ i$ are related to each other with propositional equality. From this proof we cannot tell whether some $D\ i$ is inhabited or empty.

The optimisation given for concretely collapsible families need not worry about this. In that case we have a lot more information to work with. We only have to worry about well-typed calls to the elimination operator, so we do not have to deal with deciding whether $D\ i$ is empty or not. Apart from this we only need to recover the non-recursive parts of the erased, canonical term.

If we extend the definition of internal collapsibility with something that decides whether $A\ i$ is empty or not, we get the following definition:

$$\begin{aligned} & \text{isInternallyCollapsibleDecidable} : (I : \text{Type}) (A : I \rightarrow \text{Type}) \rightarrow \text{Type} \\ & \text{isInternallyCollapsibleDecidable } I\ A = (i : I) \\ & \rightarrow (((x\ y : A\ i) \rightarrow x \equiv y) \otimes (A\ i \oplus A\ i \rightarrow \perp)) \end{aligned}$$

If we then replace the occurrence of *isInternallyCollapsible* in the type signature of *optimiseFunction* with *isInternallyCollapsibleDecidable*

3.6.1 Time complexity issues

Using this definition we do get enough information to write *optimiseFunction*. However, the success of the optimistically named function *optimiseFunction* relies on time complexity the proof given of *isInternallyCollapsibleDecidable* $D\ I$ that is used to recover the erased $A\ i$ value from the index i . In the case of concrete collapsibility this was not that much of an issue, since the way we retrieve the erased values from the indices was constant in the size of the given indices.

Apart from requiring a decision procedure that gives us, for every index $i : I$, an inhabitant of $A\ i$ or a proof that $A\ i$ is empty, we need a bound on the time complexity of this procedure. One approach, taken in Danielsson (2008) to prove time complexities of functions, is to write the functions with a monad that keeps track of how many “ticks” are needed to evaluate the function for the given input, called the *Thunk* monad. $\text{Thunk} : \mathbb{N} \rightarrow \text{Type} \rightarrow \text{Type}$ is implemented as an abstract type that comes with the following primitives:

- $\text{step} : (a : \text{Type}) \rightarrow (n : \mathbb{N}) \rightarrow \text{Thunk } n\ a \rightarrow \text{Thunk } (n + 1)\ a$
- $\text{return} : (a : \text{Type}) \rightarrow (n : \mathbb{N}) \rightarrow a \rightarrow \text{Thunk } n\ a$
- $(\gg) : (a\ b : \text{Type}) \rightarrow (n\ m : \mathbb{N}) \rightarrow \text{Thunk } m\ a \rightarrow (a \rightarrow \text{Thunk } n\ b) \rightarrow \text{Thunk } (m + n)\ b$
- $\text{force} : (a : \text{Type}) \rightarrow (n : \mathbb{N}) \rightarrow \text{Thunk } n\ a$

The user has to write its programs using these primitives. A similar approach has also been used by van Laarhoven⁵ to count the number of comparisons needed for various comparison-based sorting algorithms.

Using this to enforce a time bound on the decision procedure is not too trivial. We first need to establish what kind of time limit we want: do we want a constant time complexity, as we have with the concrete collapsibility optimisation? If we want it to be non-constant, on what variable do we want it to depend?

⁵<http://twanvl.nl/blog/agda/sorting>

Apart from these questions, approaches such as the *Thunk* monad, are prone to “cheating”: we can just write our decision procedure the normal way and then write `return 1 decisionProcedure` to make sure it has the right type. To prevent this, we can extend the list of primitives in such a way, that the users can write the program completely in this language. Such a language, if it is complete enough, will most likely make writing programs unnecessarily complex for the user.

Even though we can internalise certain conditions under which certain transformations are safe (preserve definitional equality), along with the transformations, guaranteeing that this transformation actually improves complexity proves to be a lot more difficult.

3.7 Indexed *h*-propositions and homotopy type theory

In section ?? we have seen that *h*-propositions are exactly those types that obey proof irrelevance. If we generalise this internal notion to the indexed case we arrive at something we previously have called internal collapsibility. We have also seen that if we restrict ourselves to the empty context, internal collapsibility implies collapsibility. In homotopy type theory, we are interested in postulating extra equalities needed to talk about univalence or higher inductive types. To stress the difference in what contexts we are considering, we will talk about internal collapsible for the empty context case and indexed *h*-propositions in the other case. In this section we will investigate what these differences mean when trying to optimise our programs.

When postulating extra propositional equalities, we obviously lose the canonicity property, hence we can no longer say that propositional equality implies definitional equality at run-time. The essence of the concrete collapsibility optimisation is that we need not store certain parts of our programs, because we know that they are unique, canonical and can be recovered from other parts of our program. In homotopy type theory we no longer have this canonicity property and may have to make choice in what inhabitant we recover from the indices. As an example of this we will compare two non-indexed types: the unit type and the interval. Both types are *h*-propositions, so they admit proof irrelevance, but the interval does have two canonical inhabitants that can be distinguished by definitional equality.

```
data I : Set where
  zero : Interval
  one  : Interval
  segment : zero ≡ one
```

The elimination operator for this type is defined in this way:

```
I-elim : (B : I → Type)
  → (b0 : B zero)
  → (b1 : B one)
```

$$\begin{aligned} &\rightarrow (p : (\text{transport } B \text{ segment } b_0) \equiv b_1) \\ &\rightarrow (i : I) \rightarrow B \ i \end{aligned}$$

with computation rules⁶:

$$\begin{aligned} I\text{-elim } B \ b_0 \ b_1 \ p \ \text{zero} &\triangleq b_0 \\ I\text{-elim } B \ b_0 \ b_1 \ p \ \text{one} &\triangleq b_1 \end{aligned}$$

In other words, in order to eliminate a value in the interval, we need to tell what has to be done with the endpoints interval and then have to show that this is done in such a way that the path between the endpoints is preserved.

Let us compare the above to the elimination operator for the unit type, \top :

$$\begin{aligned} \top\text{-elim} : (B : \top \rightarrow \text{Type}) \\ &\rightarrow (b : B \ tt) \\ &\rightarrow (t : \top) \rightarrow B \ t \end{aligned}$$

with computation rule:

$$\top\text{-elim } B \ b \ tt \triangleq b$$

If we have canonicity, we can clearly assume every inhabitant of \top to be tt at run-time and erase the t argument from $\top\text{-elim}$. In the case of I , we cannot do this: we have two canonical inhabitants that are propositionally equal, but not definitionally.

Not all is lost, if we consider the non-dependent elimination operator for the interval:

$$\begin{aligned} I\text{-elim-nondep} : (B : \text{Type}) \\ &\rightarrow (b_0 : B) \\ &\rightarrow (b_1 : B) \\ &\rightarrow (p : b_0 \equiv b_1) \\ &\rightarrow I \rightarrow B \end{aligned}$$

then it is easy to see that all such functions are constant functions, with respect to propositional equality. If we erase the I argument and presuppose it to be zero , we will get a new function that is propositionally equal to the original one. However, it is definitional equality that we are after. We can define the following two functions:

$$\begin{aligned} I\text{-id} : I &\rightarrow I \\ I\text{-id} &= I\text{-elim-nondep } I \ \text{zero } \text{one } \text{segment} \\ I\text{-const-zero} : I &\rightarrow I \\ I\text{-const-zero} &= I\text{-elim-nondep } I \ \text{zero } \text{zero } \text{refl} \end{aligned}$$

⁶Apart from giving computation rules for the points, we also need to give a computation rule for the path constructor, *segment*, but as we do not need this rule for the discussion here, we have left it out.

If we presuppose and erase the I argument to be *zero* in the I -*id* case, we would get definitionally different behaviour. In the case of I -*const-zero*, it does not matter if we presuppose the argument to be *zero* or *one*, since this function is also definitionally constant. This is because for the *refl* to type check, b_0 and b_1 have to definitionally equal. So if we want to optimise the elimination operators of higher inductive types that are h -propositions, such as the interval, we need to look at what paths the non-trivial paths are mapped to. If these are all mapped to *refl*, then the points all get mapped to definitionally equal points. Checking such a property can become difficult, as we can tell from this rather silly example:

```
data N-truncated : Type where
  0 : N-truncated
  S : (n : N-truncated) → N-truncated
  equalTo0 : (n : N-truncated) → 0 ≡ n
```

with non-dependent eliminator:

```
N-truncated-elim-nondep : (B : Type)
  → (b0 : B)
  → (bS : B → B)
  → (p : (b : B) → b0 ≡ b)
  → N-truncated → B
```

If we were to check that all paths between 0 and n are mapped to a *refl* between inhabitants of B , we have to check that p satisfies this property, which we cannot do.

3.7.1 Internally optimising h -propositions

The optimisation given in section 3.6 of course still is a valid transformation for the homotopy type theory case. The proof of a family $D : I \rightarrow \text{Type}$ being an indexed h -proposition is again not enough for us to be able to write the *optimiseFunction* term. What we called *isInternallyCollapsibleDecidable* is that we internally need a witness of the fact that every h -proposition in the family is either contractible or empty, so we could have written the property as follows:

```
isIndexedhPropDecidable : (I : Type) (A : I → Type) → Type
isIndexedhPropDecidable I A = (i : I)
  → (isContractible (A i)) ⊕ (A i → ⊥)
```

3.8 Conclusion and future work

Chapter 4

Applications of homotopy type theory

Bibliography

- Yves Bertot and Pierre Castéran. *Interactive theorem proving and program development: Coq'Art: the calculus of inductive constructions*. Springer-Verlag New York Incorporated, 2004.
- Ana Bove and Venanzio Capretta. Modelling general recursion in type theory. *Mathematical Structures in Computer Science*, 15(4):671–708, 2005.
- Edwin Brady, Conor McBride, and James McKinna. Inductive families need not store their indices. In *Types for Proofs and Programs*, pages 115–129. Springer, 2004.
- RL Constable, SF Allen, HM Bromley, WR Cleaveland, JF Cremer, RW Harper, DJ Howe, TB Knoblock, NP Mendler, P Panangaden, et al. *Implementing mathematics with the Nuprl proof development system*. Prentice Hall, 1986.
- Thierry Coquand. Pattern matching with dependent types. In *Informal proceedings of Logical Frameworks*, volume 92, pages 66–79, 1992.
- Nils Anders Danielsson. Lightweight semiformal time complexity analysis for purely functional data structures. In *ACM SIGPLAN Notices*, volume 43, pages 133–144. ACM, 2008.
- Pierre Letouzey. A new extraction for Coq. In *Types for proofs and programs*, pages 200–219. Springer, 2003.